
Check Please: An Automated Approach to Log Checking 
Richann Watson, Experis, Batavia, OH 

ABSTRACT 
In the Pharmaceutical industry, we find ourselves having to re-run our programs repeatedly for each deliverable.  
These programs can be run individually in an interactive SAS® session which will allow us to review the logs as we 
execute the programs.  We could run the individual program in batch and open each individual log to review for 
unwanted log messages, such as ERROR, WARNING, uninitialized, have been converted to, etc.  Both of these 
approaches are fine if there are only a handful of programs to execute.  But what do you do if you have hundreds of 
programs that need to be re-run?  Do you want to open every single one of the programs and search for unwanted 
messages?  This manual approach could take hours and is prone to accidental oversight.   This paper will discuss a 
macro that will search through a specified directory and check either all the logs in the directory or check only logs 
with a specific naming convention or check only the files listed.  The macro then produces a report that lists all the 
files checked and indicates whether or not issues were found. 

INTRODUCTION 
Checking the logs for unwanted messages is one of the first steps in ensuring that a program was executed 
successfully.  When running the program interactively this can be easily done by manually reviewing the log in the 
interactive session.  However, if you execute the program in batch mode or from a ‘driver’ program and the logs are 
saved as individual files, then you need to open the log using SAS Universal Viewer or some other type of text editor 
in order to review the log. Regardless of the method you use, the log still needs to be perused in order to find these 
pesky messages. 

TYPES OF LOG MESSAGES 
There are a variety of log messages that may cause problems in the program and these messages need to be 
investigated.  Some of the more common messages include: 

• ERROR 

• WARNING 

• UNINITIALIZED 

• MORE THAN ONE DATA SET WITH REPEATS OF BY 

• VALUES HAVE BEEN CONVERTED 

• MISSING VALUES WERE GENERATED AS A RESULT 

• INVALID DATA 

• AT LEAST ONE W.D FORMAT TOO SMALL 

How these log messages are to be fixed is dependent on the data and project needs.  ‘ERROR’ messages should 
always be corrected since this can affect the output.  Additionally, the ‘MORE THAN ONE DATA SET WITH 
REPEATS OF BY’ should be fixed so that a many-to-many merge is not done in a data step. Some other messages 
may not have an effect on the data and may not be as critical to fix; however, that is dependent on what the client 
deems as critical. There are some clients that do not wish to see any unwanted log messages and request that they 
all be fixed. 

In addition to the common messages that you may look for in the logs, the client may have a more extensive list of 
messages that they may not want to appear.  Other types of log messages that may cause issues are: 

• ORDERING BY AN ITEM THAT DOESN’T APPEAR IN 

• OUTSIDE THE AXIS RANGE 

• RETURNING PREMATURELY 

• QUERY DATA 

• QUESTIONABLE  

OTHER LOG MESSAGES TO CONSIDER 
When determining the list of possible unwanted log messages to build into the macro (provided in Appendix A), you 
may need to consider other types of messages that may cause the macro to flag the program as containing unwanted 
messages.  Additionally, if you are aware of specific types of log messages that will trigger an issue to be reported, 
then it would be wise to build into the macro the types of messages that are allowed. 

For example, when a SETINIT may be close to the expiration date, the log will have several types of WARNING 
messages. Since these WARNING messages do not have an impact on the actual execution of the program, then 
logic can be incorporated to look for these types of messages. 

• UNABLE TO COPY SASUSER 

• BASE PRODUCT PRODUCT 

• EXPIRE WITHIN 

• BASE SAS SOFTWARE …  EXPIRING SOON 

• UPCOMING EXPIRATION 

• SCHEDULED TO EXPIRE 

• SETINIT TO OBTAIN MORE INFO 

LOG CHECKING PROBLEM 
With all the various types of messages that need to be checked during the log review, one can plainly see how 
quickly this process of checks and fixes can become tedious and burdensome.  This manual approach to reviewing 
these logs can lead to several other concerns. 

1. It is easy to overlook a particular message in a log while scrolling through the log. 

2. When searching for a particular message, it is easy to mistype the message or get the casing incorrect and thus 
overlooking the unwanted message. 

3. If there are a lot of logs to check, it is easy to overlook a log altogether. 

4. The programmer may be short on time and may just make the assumption that the programs ran clean the last 
time and there is an output produced so it should still be fine. 

THE SOLUTION - CHECKLOGS 
Rather than leave the log checking to ‘chance’ and human error an automated alternative can be implemented 
through use of the CHECKLOGS macro (Appendix A).  The macro will parse through each of the indicated logs in a 
specified directory looking for the unwanted messages, and then produce a report summarizing the findings.   

MACRO PARAMETERS 
The macro has three input parameters, with only one being required: 

• ‘loc’ – location of where the log files reside. 

• ‘fnm’ – optional parameter that indicates which types of files to look at; if more than one type of file is specified 
then it should be separated by a space. 

• ‘out’ – optional parameter that indicates the name of the output report file that will be produced. 

For the optional parameter ‘fnm’, the macro will build a where clause that will be used to exclusively extract the logs 
of interest.  For example, if all your logs are stored in the same folder and you are only concerned with viewing the 
logs for the tables and figures, and all the tables and figures logs have a standard naming convention (e.g., 
‘t_#_#_#.log’ for tables and ‘f_#_#_#.log’ for figures where ‘#_#_#’ represents the output number) then you can use 
the standard portion for each type separated by a space.  

fnm = t_ l_ 

Note that the building of the where clause does not care if the standard portion is a prefix or a suffix as long as there 
is a standard and that the standard  is not used for other log types.  So if the table logs had a naming convention of 
‘xxxxx_t.log’ and the figure logs had a naming convention of ‘xxxxx_f.log’ where ‘xxxxx’ represents the output name, 
then you can still use the standard portion for each type separated by a space. 

fnm = _t. _l. 



3 

If the ‘out’ parameter is not specified, then the file name for the report will default to ‘all_checklogs.rtf’.  Ideally, if you 
are executing the macro to look at only specific types of files, then the name of the report should be specified so that 
the report is not overwritten when executed later for another file type. 

Table 1 illustrates some sample calls of the macro and what the expected outcome is.  Note that the samples in this 
paper are based off of the Windows environment.  However, the program will work with a Unix environment as well.  
The user just needs to make sure the slashes in the ‘loc’ parameter are pointing in the correct direction.  

Sample Call Expected Outcome 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix) 

Check ALL the logs in the specified directory and use 
default name “all_checklogs” for the report name. 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix, 
out=all_checklogs_20160708) 

Check ALL the logs in the specified directory and use 
specified name”all_checklogs_20160708” for the report 
name. 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix, fnm=ad_, 
out=ad_checklogs) 

Check ONLY the logs that contain “ad_” in the file name 
and use specified name “ad_checklogs” for the report 
name. 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix, fnm=sd_ ad_, 
out=sd_ad_checklogs) 

Check ONLY the logs that contain “ad_” OR “sd_” in the file 
name and use specified name “sd_ad_checklogs” for the 
report name. 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Suffix, fnm=_ad, 
out=ad_suff_checklogs) 

Check ONLY the logs that contain “_ad” in the file name 
and use specified name “ad_suff_checklogs” for the report 
name. 

Note that this example is to illustrate that the macro will still 
find the indicated logs even if the standard naming 
convention is a suffix and not a prefix. 

Table 1 Sample CHECKLOG Calls and Expected Outcome 

NUTS AND BOLTS OF THE MACRO 
So how does this macro actually work?  There are a variety of steps in the macro and each step will be explained. 

Determining Working Environment 

The first thing the macro does is determine in what working environment the command is being executed.  The macro 
is able to run in either Windows or Unix environment.  Depending on the environment, the macro will determine what 
form the pipe command should take and in which direction the slash in the path name should be pointing. 

During this initial part of the macro, the following two macro variables are set and will be used in the macro: 

• ppmcd – represents the pipe command 
• slash 

If the execution environment is Windows (i.e., &SYSSCP = WIN), then following macro variables are set as  

    %let ppcmd = %str(dir); 
    %let slash = \; 
 

If the execution environment is Unix (i.e., &SYSSCP = LIN X64), then following macro variables are set as  

    %let ppcmd = %str(ls -l); 
    %let slash = /; 

What Log Files Are Being Checked? 

If the ‘fnm’ input parameter is specified, the macro will then parse through each token in the parameter to build a 
where clause.  That where clause will use the SAS function INDEX to look for a specific value in the filename.  If no 
input parameter is specified, then the macro will check every log file in the specified directory.   

Using the sample calls in Table 1 the associated where clauses for each call are illustrated (see Table 2). 

Sample Call Where Clause Built 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix) 

No where clause.  Macro will check all logs in the directory. 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix, 
out=all_checklogs_20160708) 

No where clause.  Macro will check all logs in the directory. 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix, fnm=ad_, 
out=ad_checklogs) 

where index(flog, “ad_”) 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Prefix, fnm=sd_ ad_, 
out=sd_ad_checklogs) 

where index(flog, “sd_”) or index(flog, “ad_”) 

%checklogs(loc=C:\Users\richann\Desktop\
WUSS\Logs\Suffix, fnm=_ad, 
out=ad_suff_checklogs) 

where index(flog, “_ad”) 

Table 2 Sample CHECKLOG Calls and Where Clause Built 

Extracting the Logs 

After the pipe command is built and the where clause is determined (if applicable) the program will bring in every file 
within the specified directory and create a data set.  This data set will only contain the logs files that are indicated by 
the where clause, in the event that a where clause is built.  Additionally, the name of the log, the date and time can be 
extracted from the various tokens in the filename.  When the pipe command reads in each file all the information for 
that file is captured on one line (one variable) and the data can be parsed to extract each component.   

Table 3 below illustrates how the filename would look if executed in a Windows environment and what the log, date 
and time are once they are extracted. 

filename flog ftim fdat 
09/28/2015 05:37 AM 32,668 ad_adae.log ad_adae 05:37 AM 28SEP2015 
09/28/2015 05:37 AM 27,166 ad_adex.log ad_adex 05:37 AM 28SEP2015 
09/30/2015 10:01 AM 40,178 ad_adlb.log ad_adlb 10:01 AM 30SEP2015 
09/22/2015 05:19 AM 41,698 ad_adsl.log ad_adsl 05:19 AM 22SEP2015 
08/28/2015 06:14 AM 20,028 sd_AE.log sd_AE 06:14 AM 28AUG2015 
09/01/2015 05:41 AM 22,392 sd_DA.log sd_DA 05:41 AM 01SEP2015 
08/26/2015 11:59 AM 19,734 sd_dm.log sd_dm 11:59 AM 26AUG2015 
09/07/2015 03:43 AM 18,615 sd_DS.log sd_DS 03:43 AM 07SEP2015 
08/26/2015 12:00 PM 18,452 sd_EX.log sd_EX 12:00 PM 26AUG2015 
09/11/2015 09:06 AM 46,331 sd_LB.log sd_LB 09:06 AM 11SEP2015 
09/02/2015 09:12 AM 28,045 sd_PE.log sd_PE 09:12 AM 02SEP2015 

Table 3 Files Retrieved and Log Name, Date and Time Extracted 

Parsing Each Log File and Creating the Log Report 

Once it is determined which logs are to be checked, the macro will then go through each file and check for ‘common’ 
undesired log messages as well as other ‘not-so-common’ undesired log messages.  At this point, the user may wish 
to add ‘user-defined’ log messages that should be investigated.  If an undesired log message is found, it is saved to a 
data set so it can be included in the check log report.  If there are no undesired log messages, then a ‘dummy’ record 
is created for that file with a generic message so that the user can be confident that the log was actually checked and 
not inadvertently left out of the checking process. 

After each log is checked a report will be generated.  If the ‘out’ parameter is specified during the macro call, the 
value of the parameter will be used as the name of the report otherwise the name of the report will default to 
‘all_checklogs’.  The report will contain the name of the log, the date the log was created, the time it was created (if 
applicable) and the undesired log message or generic message.  In addition to this, the report will contain a blank 

column that will allow the user to input any additional information such as a comment as to why that specific message 
was allowed while they are reviewing the report.   

For example, in Display 1 the log SD_DM.log had a message that contained the word ‘warning’.  This was part of a 
comment so was not an actual program warning, thus the message is allowed and so a reason is manually added to 
the report during the review process.  Furthermore, the log for SD_LB had three messages that contained ‘??’ (a 
sponsor-defined message).  For one of the messages this was part of the comment so the message is allowed and a 
reason is manually added.  For the other two messages, a comment is added indicating that for this particular 
deliverable it will be allowed but must be fixed for the next deliverable. 

 
Display 1 Sample CHECKLOGS Report 

CONCLUSION 
Checking the logs for a deliverable is a task we all dread doing, but it must be done. It is a necessary ‘evil’.  However, 
with the CHECKLOGS macro, this process has been simplified and produces a report that can be saved with the 
deliverable, and also help explain any log messages that were not addressed.  

REFERENCES 
UNIX commands 

http://www.shareittips.com/data/unix.png 

CMD commands 

http://ss64.com/nt/ 

ACKNOWLEDGMENTS 
Thanks to Karl Miller and Deanna Schreiber-Gregory for reviewing the paper and providing additional insight. 

Log Name Log Date Log Time Log Message 
Reason Log Message is 

Allowed 
AD_ADAE 28SEP2015 05:37 AM No undesired messages.  Log is clean.  

 

AD_ADEX 28SEP2015 05:37 AM No undesired messages.  Log is clean.  

 

AD_ADLB 30SEP2015 10:01 AM No undesired messages.  Log is clean.  

 

AD_ADSL 22SEP2015 05:19 AM No undesired messages.  Log is clean.  

 

SD_AE 28AUG2015 06:14 AM No undesired messages.  Log is clean.  

 

SD_DA 01SEP2015 05:41 AM NOTE: Character values have been converted to 
numeric values at the places given by: 

 

 01SEP2015 05:41 AM NOTE: Character values have been converted to 
numeric values at the places given by: 

 

 01SEP2015 05:41 AM NOTE: Missing values were generated as a result of 
performing an operation on missing values. 

 

 

SD_DM 26AUG2015 11:59 AM 23         SITEID at the begining of the first data step to 
avoid warning messages. A worning 

This is a comment.  There is no 
issue to fix. 

 

SD_EX 26AUG2015 12:00 PM No undesired messages.  Log is clean.  

 

SD_LB 11SEP2015 09:06 AM 402      ! ??)*/ This is a comment.  There is no 
issue to fix. 

 11SEP2015 09:06 AM 403          RESLTVALN = input(RESLTVAL, 
?? best12.); 

‘??’ is used as part of the format 
statement.  To be fixed in next 
deliverable. 

 11SEP2015 09:06 AM 407           RESLTVALN = 
input(substr(RESLTVAL,2,length(RESLTVAL)-1), ?? 
best12.); 

‘??’ is used as part of the format 
statement.  To be fixed in next 
deliverable. 

 

SD_PE 02SEP2015 09:12 AM NOTE: Character values have been converted to 
numeric values at the places given by: 

 

http://www.shareittips.com/data/unix.png
http://ss64.com/nt/

Institute Inc. in the USA and other countries. ® indicates USA registration.  

APPENDIX A: CHECKLOGS MACRO 
/* retrieve all the logs in the specified directory */ 
%macro checklogs(loc=, /* location of where the log files are stored       */ 
                 fnm=, /* which types of files to look at (optional)       */ 
            /* e.g., Tables – t_, Figures – f_, Listings – l_   */  
      /* separate types of files by a space (e.g., t_ f_) */ 
                 out=  /* log report name (optional)                       */ ); 
     
  /* need to determine the environment in which this is executed   */ 
  /* syntax for some commands vary from environment to environment */ 
  %if &sysscp = WIN %then %do; 
    %let ppcmd = %str(dir); 
    %let slash = \; 
  %end; 
  %else %if &sysscp = LIN X64 %then %do; 
    %let ppcmd = %str(ls -l); 
    %let slash = /; 
  %end; 
 
  /* if a filename is specified then build the where clause */ 
  %if &fnm ne   %then %do; 
    data _null_; 
      length fullwhr $2000.; 
      retain fullwhr; 
 
      /* read in each log file and check for undesired messages */ 
      %let f = 1; 
      %let typ = %scan(&fnm, &f); 
 
      /* loop through each type of filename to build the where clause */ 
      %do %while (&typ ne ); 
 
   partwhr = catt("index(flog, '", "&typ", "')"); 
   fullwhr = catx(" or ", fullwhr, partwhr); 
 
   call symputx('fullwhr', fullwhr); 
 
         %let f = %eval(&f + 1); 
   %let typ = %scan(&fnm, &f); 
      %end; 
 
    run; 
  %end; 
 
  /* need to build pipe directory statement as a macro var  */ 
  /* because the statement requires a series of single and  */ 
  /* double quotes - by building the directory statement    */ 
  /* this allows the user to determine the directory rather */ 
  /* than it being hardcoded into the program               */ 
  /* macro var will be of the form:'dir "directory path" '  */ 

mailto:richann.watson@experis.com

  data _null_; 
    libnm = "&loc"; 
    dirnm = catx(" ", "'", "&ppcmd", quote(libnm), "'"); 
    call symputx('dirnm', dirnm); 
  run; 
 
  /* read in the contents of the directory containing the logs */ 
  filename pdir pipe &dirnm lrecl=32727; 
 
  data logs (keep = flog fdat ftim); 
    infile pdir truncover scanover; 
    input filename $char1000.; 
    length flog $25 fdat ftim $10; 
 
    /* keep only the logs */ 
    if index(filename, ".log"); 
 
    /* parse out the string to get the log name */ 
    /* note on the scan function a negative #   */ 
    /* scans from the right and a positive #    */ 
    /* scans from the left                      */ 
    /* for log keep only first part (e.g. 'ae') */ 
    flog = scan(scan(filename, -1, " "), 1); 
 
    /*********** WINDOWS ENVIRONMENT ************/ 
    /* the pipe will read in the information in */ 
    /* the format of: date time am/pm size file */ 
    /* e.g. 08/24/2015 09:08 PM 18,498 ae.log   */ 
    /*    '08/24/2015' is first token from left */ 
    /*    'ae.log' is first token from right    */ 
    %if &sysscp = WIN %then %do; 
      ftim = catx(" ", scan(filename, 2, " "), scan(filename, 3, " ")); 
      fdat = put(input(scan(filename, 1, " "), mmddyy10.), date9.); 
    %end; 
 
    /***************************** UNIX ENVIRONMENT ******************************/ 
    /* the pipe will read in the information in the format of: permissions, user,*/ 
    /* system environment, file size, month, day, year or time, filename         */ 
    /* e.g. -rw-rw-r-- 1 userid sysenviron 42,341 Oct 22 2015 ad_adaapasi.log    */ 
    /*    '-rw-rw-r--' is first token from left                                  */ 
    /*    'ad_adaapasi.log' is first token from right                            */ 
    %else %if &sysscp = LIN X64 %then %do; 
      _ftim = scan(filename, 8, " "); 
 
      /* in Unix if year is current year then time stamp is displayed */ 
      /* otherwise the year last modified is displayed                */ 
      /* so if no year is provided then default to today's year and if*/ 
      /* no time is provided indicated 'N/A'                          */ 
      if anypunct(_ftim) then do; 
        ftim = put(input(_ftim, time5.), timeampm8.); 
        yr = put(year(today()), Z4.); 
      end; 
      else do; 
        ftim = 'N/A'; 
        yr = _ftim; 
      end; 
  
      fdat = cats(scan(filename, 7, " "), upcase(scan(filename, 6, " ")), yr); 
    %end; 
  run; 

  /* create a list of logs, dates, times and store in macro variables */ 
  proc sql noprint; 
    select flog, 
           fdat,  
           ftim 
           into : currlogs separated by " ", 
          : currdats separated by " ", 
          : currtims separated by "@" 
    from logs 
    %if &fnm ne   %then where &fullwhr;     ; /* need to keep extra semicolon */ 
  quit; 
  /* need to make sure the alllogs data set does not exist before getting into loop */ 
  proc datasets; 
    delete alllogs; 
  quit; 
 
  /* read in each log file and check for undesired messages */ 
  %let x = 1; 
  %let lg = %scan(&currlogs, &x); 
  %let dt = %scan(&currdats, &x); 
  %let tm = %scan(&currtims, &x, '@'); 
 
  /* loop through each log in the directory and look for undesirable messages */ 
  %do %while (&lg ne ); 
    /* read the log file into a SAS data set to parse the text */ 
    data logck&x; 
      infile "&loc.&slash.&lg..log" missover pad; 
      input line $1000.; 
 
      /* keep only the records that had an undesirable message */ 
      if index(upcase(line), "WARNING") or 

  index(upcase(line), "ERROR:") or 
  index(upcase(line), "UNINITIALIZED") or 
  index(upcase(line), "NOTE: MERGE") or 
  index(upcase(line), "MORE THAN ONE DATA SET WITH REPEATS OF BY") or 
  index(upcase(line), "VALUES HAVE BEEN CONVERTED") or 
  index(upcase(line), "MISSING VALUES WERE GENERATED AS A RESULT") or 
  index(upcase(line), "INVALID DATA") or 
  index(upcase(line), "INVALID NUMERIC DATA") or 
  index(upcase(line), "AT LEAST ONE W.D FORMAT TOO SMALL") or 
  index(upcase(line), "ORDERING BY AN ITEM THAT DOESN'T APPEAR IN") or 
  index(upcase(line), "OUTSIDE THE AXIS RANGE") or 
  index(upcase(line), "RETURNING PREMATURELY") or 
  index(upcase(line), "UNKNOWN MONTH FOR") or 
  index(upcase(line), "QUERY DATA") or 
  index(upcase(line), "??") or 
  index(upcase(line), "QUESTIONABLE"); 
 

      /* create variables that will contain the log that is being scanned */ 
      /* as well as the and date and time that the log file was created   */ 
      length lognm $25. logdt logtm $10.; 
      lognm = upcase("&lg"); 
      logdt = "&dt"; 
      logtm = "&tm"; 
 
      /* create a dummy variable to create a column on the report that will allow */ 
      /* users to enter a reason if the message is allowed                        */ 
      logrs = ' '; 
    run; 
 
     
  



Check Please: An Automated Approach to Log Checking, continued 

 

9 

    /* because there are sometimes issues with SAS certificate */ 
    /* there will be warnings in the logs that are expected    */ 
    /* these need to be removed                                */ 
    data logck&x._2; 
      set logck&x.; 
      if index(upcase(line), 'UNABLE TO COPY SASUSER') or 
         index(upcase(line), 'BASE PRODUCT PRODUCT') or 
         index(upcase(line), 'EXPIRE WITHIN') or 
         (index(upcase(line), 'BASE SAS SOFTWARE') and  
          index(upcase(line), 'EXPIRING SOON')) or 
         index(upcase(line), 'UPCOMING EXPIRATION') or 
         index(upcase(line), 'SCHEDULED TO EXPIRE') or 
         index(upcase(line), 'SETINIT TO OBTAIN MORE INFO') then delete; 
    run; 
    /* determine the number of undesired messages were in the log */ 
    data _null_; 
      if 0 then set logck&x._2 nobs=final; 
      call symputx('numobs',left(put(final, 8.))); 
    run; 
 
    /* if there is no undesired messages in log create a dummy record for report */ 
    %if &numobs = 0 %then %do; 
      data logck&x._2; 
        length lognm $25. line $1000. logdt logtm $10.; 
        line = "No undesired messages.  Log is clean."; 
        lognm = upcase("&lg"); 
        logdt = "&dt"; 
        logtm = "&tm"; 
 
  /* create a dummy variable to create a column on the report that will allow */ 
  /* users to enter a reason if the message is allowed                        */ 
  logrs = ' '; 
  output; 
      run; 
    %end; 
 
    /* append all the results into one data set */ 
    %if &x = 1 %then %do; 
      data alllogs; 
        set logck&x._2; 
      run; 
    %end; 
    %else %do; 
      proc append base=alllogs 
                  new=logck&x._2; 
      run; 
    %end; 
 
    %let x = %eval(&x + 1); 
    %let lg = %scan(&currlogs, &x); 
    %let dt = %scan(&currdats, &x); 
    %let tm = %scan(&currtims, &x, '@'); 
  %end; 
 
  /* since a list of files can be provided then the files may not be in order */ 
  proc sort data=alllogs presorted; 
    by lognm line; 
  run; 
 
  /* if the name of the output file is not specified then default to the name */ 
  %if &out =  %then %do; 
    %let out=all_checklogs; 
  %end; 

  /* create the report */ 
  ods listing close; 
  options orientation=landscape; 
 
  ods rtf file="&loc.&slash.&out..rtf"; 
 
  proc report data=alllogs ls=140 ps=43 spacing=1 missing nowindows headline; 
    column lognm logdt logtm line logrs;  
    define lognm / order   style(column)=[width=12%]      "Log Name";  
    define logdt / display style(column)=[width=12%]      "Log Date";  
    define logtm / display style(column)=[width=12%]      "Log Time";  
    define line  / display style(column)=[width=30%] flow "Log Message"; 
    define logrs / display style(column)=[width=20%] flow "Reason Message is Allowed"; 
 
    /* force a blank line after each file */ 
    compute after lognm; 
      line " "; 
    endcomp; 
  run;  
 
  ods rtf close; 
  ods listing; 
%mend checklogs; 

